

STACK

$\text{STACK [TOP] } \left\{ \begin{array}{l} \text{COMMAND } \textit{operand1} [\textit{operand2} [(parameter)]] \dots \\ \text{[DATA] [FORMATTED] } \{ \textit{operand2} [(parameter)]] \dots \end{array} \right\}$

This chapter covers the following topics:

- Function
- Syntax Description
- Example

For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

Related Statements: INPUT | RELEASE

Function

The STACK statement is used to place any of the following into the Natural stack:

- the name of a Natural program or Natural system command to be executed;
- data to be used during the execution of an INPUT statement.

For further information on the stack, see *Further Programming Aspects, Stack* in the *Programming Guide*.

Syntax Description

Operand Definition Table:

Operand	Possible Structure					Possible Formats										Referencing Permitted	Dynamic Definition				
<i>operand1</i>	C	S	A	G	N	A														yes	yes
<i>operand2</i>	C	S	A	G	N	A	U	N	P	I	F	B	D	T	L	G	yes	yes			

Syntax Element Description:

Syntax Element	Description
TOP	<p>TOP Option:</p> <p>If you specify TOP, the data/program/command will be placed at the top of the Natural stack. Otherwise, they are placed at the bottom of the stack.</p> <p>Example: The following statement causes the content of the variable #FIELDA to be placed as data on top of the stack:</p> <pre>STACK TOP #FIELDA</pre>
DATA	<p>DATA Option:</p> <p>This option, which is also the default, causes data to be placed in the stack which are to be used as input data for an INPUT statement.</p> <p>Delimiter characters or input assign characters contained within the data values will be processed as delimiters. For details on how data from the stack are processed by an INPUT statement, refer to <i>Processing Data from the Natural Stack</i> (in the description of the INPUT statement).</p> <p>Example: The following statements cause the contents of the variables #FIELD1 and #FIELD2 to be placed in the stack:</p> <pre>MOVE 'ABC' TO #FIELD1 MOVE 'XYZ' TO #FIELD2 STACK #FIELD1 #FIELD2</pre> <p>These variables will be passed as data to the next INPUT statement in the Natural program, using delimiter mode:</p> <pre>INPUT #FIELD1 #FIELD2</pre> <p>Note: If <i>operand2</i> is a time variable (Format T), only the time component of the variable content is placed in the stack, but not the date component.</p>

Syntax Element	Description
FORMATTED	<p>FORMATTED Option:</p> <p>This option causes all data to be passed on a field-by-field basis to the next INPUT statement; no key assignments or delimiter characters will be interpreted.</p> <p>Examples:</p> <p>The following statements cause ABC , DEF to be placed in #FIELD1 and XYZ in #FIELD2:</p> <pre>MOVE 'ABC,DEF' TO #FIELD1 MOVE 'XYZ' TO #FIELD2 STACK TOP DATA FORMATTED #FIELD1 #FIELD2 ... INPUT #FIELD1 #FIELD2</pre> <p>Assuming the input delimiter character to be the comma (profile/session parameter ID= ,), the following statements - without the keyword FORMATTED - cause ABC to be placed in #FIELD1 and DEF in #FIELD2:</p> <pre>MOVE 'ABC,DEF' TO #FIELD1 STACK TOP DATA #FIELD1 ... INPUT #FIELD1 #FIELD2</pre> <p>Note: The FORMATTED option should be used if the data to be passed contains delimiter, control or DBCS characters to avoid unintentional interpretation of these characters.</p>
COMMAND <i>operand1</i>	<p>COMMAND Option:</p> <p>To place a command (or program name) in the stack, you specify the keyword COMMAND followed by the command specified in <i>operand1</i>. Natural will execute the command instead of displaying the NEXT prompt and prompting the user for input.</p> <p>Example:</p> <p>The following statement causes the command RUN to be placed at the top of the stack. Natural will execute this command at the point where the NEXT prompt would normally be issued.</p> <pre>STACK TOP COMMAND 'RUN'</pre>

Syntax Element	Description
COMMAND <i>operand1 operand2</i> . . .	<p>COMMAND with Data Option:</p> <p>Together with a command (<i>operand1</i>), you may also place data (<i>operand2</i>) in the stack. These data will then be processed by the next INPUT statement after the command has been executed.</p> <p>Data stacked with a command are always stacked unformatted.</p> <p>Note:</p> <p>If the data to be stacked include empty alphanumeric fields (that is, blanks), these blanks will be interpreted as delimiters between values and thus not processed correctly by the corresponding INPUT statement. Therefore, if you wish to stack empty alphanumeric fields as data with a command, you have to use two STACK statements: one STACK DATA <i>operand2</i> . . . to stack the data, and one STACK COMMAND <i>operand1</i> to stack the command.</p>
<i>parameter</i>	<p>Date Format:</p> <p>If <i>operand2</i> is a date variable, you can specify the session parameter DF as a parameter for this variable.</p>

Example

```

** Example 'STKEX1': STACK
*****
DEFINE DATA LOCAL
1 #CODE (A1)
END-DEFINE
*
INPUT //
  10X 'PLEASE SELECT COMMAND' //
  10X 'LIST VIEW          (V)' /
  10X 'LIST PROGRAM * (P)' /
  10X 'TECH INFO         (T)' /
  10X 'STOP              (.)' //
  20X 'CODE:' #CODE
*
*
DECIDE ON FIRST #CODE
  VALUE 'V'
    STACK TOP DATA   'VIEW'
    STACK TOP COMMAND 'LIST'
  VALUE 'P'
    STACK TOP COMMAND 'LIST PROGRAM *'
  VALUE 'T'
    STACK TOP COMMAND 'LAST *'
    STACK TOP COMMAND 'TECH'
    STACK TOP COMMAND 'SYSPROD'
  VALUE '.'
  STOP

```

```
NONE
  REINPUT 'PLEASE ENTER VALID CODE'
END-DECIDE
*
*
END
```

Output of Program STKEX1:

PLEASE SELECT COMMAND

```
LIST VIEW      (V)
LIST PROGRAM * (P)
TECH INFO      (T)
STOP           (.)
```

CODE:P

After entering and confirming code:

```
16:46:28          ***** NATURAL LIST COMMAND *****          2005-01-19
User HTR          - LIST Objects in a Library -          Library SYSEXSYN
```

Cmd	Name	Type	S/C	SM	Version	User ID	Date	Time
---	*_____	P_____	*_	*	*_____	*_____	*_____	*_____
___	ACREX1	Program	S/C	S	4.1.03	RKE	2004-11-11	16:32:37
___	ACREX2	Program	S/C	S	4.1.03	RKE	2005-01-05	10:29:51
___	ADDEX1	Program	S/C	S	4.1.03	RKE	2004-11-11	16:36:49
___	AEDEX1R	Program	S/C	R	4.1.03	RKE	2004-11-11	16:40:34
___	AEDEX1S	Program	S/C	S	4.1.03	RKE	2004-11-11	16:39:57
___	AEPEX1R	Program	S/C	R	4.1.03	RKE	2004-11-11	16:41:57
___	AEPEX1S	Program	S/C	S	4.1.03	RKE	2004-11-11	16:42:31
___	AEPEX2	Program	S/C	S	4.1.03	RKE	2004-11-11	16:43:37
___	ASDEX1R	Program	S/C	R	4.1.03	RKE	2004-11-11	17:00:21
___	ASDEX1S	Program	S/C	S	4.1.03	RKE	2004-11-11	17:00:50
___	ASGEX1R	Program	S/C	R	4.1.03	RKE	2004-11-11	17:02:01
___	ASGEX1S	Program	S/C	S	4.1.03	RKE	2004-11-11	17:02:08
___	ATBEX1R	Program	S/C	R	4.1.03	RKE	2004-11-11	17:03:18
___	ATBEX1S	Program	S/C	S	4.1.03	RKE	2004-11-11	17:03:05

14 Objects found

Top of List.

Command ==>

```
Enter-PF1---PF2---PF3---PF4---PF5---PF6---PF7---PF8---PF9---PF10---PF11---PF12---
      Help Print Exit Sort      -- - + ++ > Canc
```